Question: How can I plot a Möbius strip? I tried [this one](http://mathworld.wolfram.com/MoebiusStrip.html), but I can't get it to work:

```
ParametricPlot3D[{
    (5 + s*Cos[u/2]) Cos[u],
    (5 + s*Cos[u/2]) Sin[u],
    (s*Sin[u/2])}, {u, -20, 20}, {s, 0, 2 π}]

```

[![my strip](https://i.stack.imgur.com/z3aQV.png)](https://i.stack.imgur.com/z3aQV.png)

Answer: 
Just for fun (using the parametrization from OP hyperlink):

```
mobius[r_, s_,t_] := {r + s Cos[t/2], r + s Cos[t/2], s Sin[t/2]} {Cos[t], Sin[t],
    1}
Manipulate[
 With[{wd = w},
  Row[{
    ParametricPlot[{u, v}, {u, -wd, wd}, {v, 0, 2 Pi}, 
     AspectRatio -> 1/2, Epilog -> {Red, PointSize[0.04], Point@p}, 
     ImageSize -> 200],
    Show[ParametricPlot3D[mobius[r, u, v], {u, -wd, wd}, {v, 0, 2 Pi},
       Mesh -> False, PlotStyle -> Yellow, 
      PerformanceGoal -> "Quality"], 
     Graphics3D[{Red, PointSize[0.04], Point[mobius[r, ##] & @@ p]}], 
     ImageSize -> 200]
    }]], {r, 1, 2}, {w, 0.5, 2}, {{p, {0, 0}}, {{-w, 0}, {w, 2 Pi}}, 
  Slider2D}]

```

[![enter image description here](https://i.stack.imgur.com/keFMO.gif)](https://i.stack.imgur.com/keFMO.gif)
...and, just for the sake of variation, here is a "minimal" Möbius strip (that is, a [minimal surface](http://mathworld.wolfram.com/MinimalSurface.html) with the topology of a Möbius strip). This example is due to [Meeks](http://dx.doi.org/10.1215/S0012-7094-81-04829-8).

```
ParametricPlot3D[{(r^3 - 1/r^3) Cos[3 θ]/3 + (r^2 + 1/r^2) Cos[2 θ] + (r - 1/r) Cos[θ],
                  (r^3 - 1/r^3) Sin[3 θ]/3 + (r^2 + 1/r^2) Sin[2 θ] + (r - 1/r) Sin[θ],
                  2 (r - 1/r) Sin[θ]}, {r, 1, 4/3}, {θ, -π, π}]

```

![minimal Möbius strip](https://i.stack.imgur.com/3a4Zq.png)